#QT += core gui opengl

TARGET = Tanks
TEMPLATE = app

# You need to have an environment variable set that points to a directory containing all the required libs
# That dir should contain libraries inside directories with the following structure:
# LIB_NAME - contains the compiled library
# LIB_NAME\LIB_NAME - contains the headers
#
# example:
# [$$LIBS_DIR]
# |- [SDL]
#    |- SDL.lib SDLmain.lib SDL.dll
#    |- [SDL]
#       |- SDL.h SDL_active.h SDL_audio.h <...>

EXTERNAL_LIBS_AND_PATHS = SDL
ADDITIONAL_LIBS = SDLmain opengl32

for(LIB, EXTERNAL_LIBS_AND_PATHS) {
    INCLUDEPATH += "$$(LIBS_DIR)\\$$LIB"
    LIBS += -L"$$(LIBS_DIR)\\$$LIB" -l$$LIB
}

for(LIB, ADDITIONAL_LIBS) {
    LIBS += -l$$LIB
}

SOURCES += main.cpp \
    tankmodel.cpp \
    terrain.cpp \
    sdlwindow.cpp \
    utils.cpp \
    projectile.cpp

HEADERS += \
    tankmodel.h \
    terrain.h \
    sdlwindow.h \
    utils.h \
    projectile.h

FORMS += \
    mainwindow.ui


















